

T-SQL:

Procedure, kursori i funkcije

Baze podataka 2 (2022/2023)

OAS Primenjeno softversko inženjerstvo

Fakultet tehničkih nauka

Univerzitet u Novom Sadu

T-SQL

- *Transact-SQL (T-SQL)*
 - proširenje klasičnog jezika *SQL*
 - podrška za brojne dodatne koncepte i naredbe
 - posebno za proceduralno programiranje u kontekstu relacionih baza podataka
 - korišćenje *T-SQL* na serveru za upravljanje bazama podataka *Microsoft SQL Server*
 - mogućnost izdavanja naredbi u *T-SQL* putem alata *Microsoft SQL Server Management Studio*

T-SQL

- *Transact-SQL (T-SQL)*
 - primeri podržanih dodatnih koncepata
 - promenljive
 - kontrola toka
 - grananje
 - ciklus
 - ...
 - procedure
 - funkcije
 - kursori
 - trigeri
 - ...

T-SQL

- Korišćeni softver u primerima i zadacima
 - *Microsoft SQL Server 2019 Express (SQL Server)*
 - *Microsoft SQL Server Management Studio v18.11.1 (SSMS)*
- Korišćeni podaci u primerima i zadacima
 - baza podataka *Geobaza*
 - **priložena datoteka *skript-tabele.sql***
 - **priložena datoteka *skript-podaci.sql***

T-SQL

- Promenljive
 - deklaracija promenljive
 - pomoću iskaza DECLARE
 - primer za promenljivu *Prom*
DECLARE @Prom integer;
 - dodela vrednosti
 - pomoću iskaza SET
 - primer za promenljivu *Prom*
SET @Prom = 1;
 - pomoću iskaza SELECT

T-SQL

- Kontrola toka

- grananje IF...ELSE

- osnovna struktura

```
IF LOGIČKI IZRAZ
```

```
    GRANA IF
```

```
ELSE
```

```
    GRANA ELSE
```

- u slučaju da logički izraz ima vrednost TRUE izvršava se grana IF, inače se izvršava grana ELSE
 - logički izraz može sadržati iskaz SELECT
 - grane mogu biti blokovi
 - konstrukcija BEGIN...END
 - deo ELSE nije obavezan

T-SQL

- Kontrola toka

- ciklus WHILE

- osnovna struktura

WHILE LOGIČKI IZRAZ
TELO

- ponavljajuće izvršavanje tela ciklusa u slučaju da logički izraz ima vrednost TRUE
 - logički izraz može sadržati iskaz SELECT
 - mogućnost dodatne kontrole ciklusa pomoću posebnih rezervisanih reči
 - BREAK za prekid ciklusa
 - CONTINUE za ponovno pokretanje ciklusa

T-SQL

- Primer za korišćenje promenljive i grananje

```
DECLARE @Sati integer;
```

```
SET @Sati = DATEPART(hour, SYSDATETIME());
```

```
IF @Sati < 12
```

```
    PRINT 'pre podneva'
```

```
ELSE IF @Sati = 12 AND DATEPART(minute, SYSDATETIME()) = 0
```

```
    PRINT 'podne'
```

```
ELSE
```

```
    PRINT 'posle podneva'
```

```
GO
```

T-SQL

- Procedure
 - rutine koje dozvoljavaju složene obrade
 - dozvoljavaju korišćenje ulaznih i izlaznih parametara
 - moguće podrazumevane vrednosti za parametre
 - skladištenje na serverskoj strani
 - uskladištene procedure
 - osnovne naredbe
 - CREATE PROCEDURE za kreiranje procedure
 - ALTER PROCEDURE za izmenu procedure
 - DROP PROCEDURE za uklanjanje procedure
 - DROP PROCEDURE IF EXISTS NAZIV PROCEDURE
 - EXEC za poziv procedure

T-SQL

- Primer procedure – pretraživanje država po nazivu (korišćenje ulaznog parametra)

```
CREATE PROCEDURE dbo.PretrazivanjeDrzave
    @NazivDrzave varchar(50)
AS
    SELECT @NazivDrzave = '%' + @NazivDrzave + '%';
    SELECT DrzOznaka, DrzNaziv, DrzGlavniGrad, UreNaziv
    FROM dbo.DRZAVA, dbo.UREDJENJE
    WHERE dbo.DRZAVA.DrzUredjenje = dbo.UREDJENJE.UreId and
           dbo.DRZAVA.DrzNaziv LIKE @NazivDrzave;
GO

EXEC dbo.PretrazivanjeDrzave @NazivDrzave = 'Koreja';
GO
```

T-SQL

- Primer procedure – određivanje prosečne dužine naziva države (korišćenje izlaznog parametra)

```
CREATE PROCEDURE dbo.PretrazivanjeDrzaveNazivDuzinaProsek
    @DuzinaNazivProsek decimal(5, 2) OUTPUT
AS
    SELECT @DuzinaNazivProsek = AVG(CAST(LEN(DrzNaziv) AS
    decimal (5,2)))
    FROM dbo.DRZAVA;
GO
```

```
DECLARE @Rezultat decimal(5, 2);
EXEC dbo.PretrazivanjeDrzaveNazivDuzinaProsek
    @DuzinaNazivProsek = @Rezultat OUTPUT;
PRINT 'Prosecna duzina naziva drzave: ' +
    CAST(@Rezultat AS VARCHAR(10));
GO
```

T-SQL

- Primer procedure – unos podataka o državi (korišćenje pomoćne promenljive)

```
CREATE PROCEDURE dbo.UpisivanjeDrzave
    @Oznaka char(3),
    @Naziv varchar(50),
    @GlavniGrad varchar(50),
    @UredjenjeOznaka char(3)
AS
    DECLARE @UredjenjeId integer;
    SELECT @UredjenjeId = UreId
    FROM dbo.UREDJENJE
    WHERE UreOznaka = @UredjenjeOznaka;
    INSERT INTO DRZAVA(DrzOznaka, DrzNaziv, DrzGlavniGrad,
        DrzUredjenje)
    VALUES (@Oznaka, @Naziv, @GlavniGrad, @UredjenjeId);
GO

EXEC dbo.UpisivanjeDrzave 'GHA', 'Gana', 'Akra', 'REP';
GO
```

T-SQL

- Zadatak za procedure
 - Napisati proceduru koja ispisuje za svaki kontinent koliko se na njemu prostire država

T-SQL

- Zadatak za procedure
 - Napisati proceduru za unos podataka o prostiranju države na kontinentu
 - na ulazu zadati oznaku kontinenta i oznaku države
 - rezultat je pokazatelj uspešnosti unosa

T-SQL

- Kursori
 - mehanizam za iteriranje kroz podatke (skup rezultata) koji dozvoljava prolazak red po red
 - osnovne radnje nad kursorom
 - deklaracija kursora (DECLARE)
 - otvaranje kursora (OPEN)
 - učitavanje sadržaja pomoću kursora (FETCH)
 - zatvaranje kursora (CLOSE)
 - dealokacija kursora (DEALLOCATE)

T-SQL

- Primer kursora – prikazivanje podataka o državama, deo 1

```
DECLARE @DrzavaOznaka char(3),  
        @DrzavaNaziv varchar(50),  
        @DrzavaGlavniGrad varchar(50),  
        @DrzavaUredjenjeId char(3),  
        @DrzavaOpis varchar(150);
```

```
DECLARE DrzaveKursor CURSOR FOR  
SELECT DrzOznaka, DrzNaziv, DrzGlavniGrad, DrzUredjenje  
FROM dbo.DRZAVA;
```

```
OPEN DrzaveKursor
```

```
FETCH NEXT FROM DrzaveKursor  
INTO @DrzavaOznaka, @DrzavaNaziv, @DrzavaGlavniGrad,  
    @DrzavaUredjenjeId
```

T-SQL

- Primer kursora – prikazivanje podataka o državama, deo 2

```
WHILE @@FETCH_STATUS = 0
BEGIN
    SELECT @DrzavaOpis = UreNaziv
    FROM dbo.UREDJENJE
    WHERE UreId = @DrzavaUredjenjeId;

    PRINT '[Kursor] ' + @DrzavaOpis + ' ' +
        @DrzavaNaziv + ' (' + @DrzavaOznaka +
        '), glavni grad ' + @DrzavaGlavniGrad;

    FETCH NEXT FROM DrzaveKursor
    INTO @DrzavaOznaka, @DrzavaNaziv, @DrzavaGlavniGrad,
        @DrzavaUredjenjeId
END

CLOSE DrzaveKursor;
DEALLOCATE DrzaveKursor;
GO
```

T-SQL

- Zadatak za kursore
 - Napisati kursor koji će za svako državno uređenje prikazati koliko ima država s takvim uređenjem

T-SQL

- Funkcije
 - rutine koje dozvoljavaju obrade
 - prihvataju parametre
 - vraćaju rezultat u skalarnom ili tabelarnom obliku
 - osnovne naredbe
 - CREATE FUNCTION za kreiranje funkcije
 - ALTER FUNCTION za izmenu funkcije
 - DROP FUNCTION za uklanjanje funkcije
 - DROP FUNCTION IF EXISTS NAZIV_FUNKCIJE

T-SQL

- Primer funkcije – formiranje pomoćne oznake za državu

```
CREATE FUNCTION dbo.FormiranjeOznake(@naziv varchar(50))
RETURNS char(3)
AS
BEGIN
    DECLARE @oznaka char(3);
    SET @oznaka = SUBSTRING(UPPER(TRIM(@naziv)), 1, 3);
    RETURN(@oznaka);
END
GO

SELECT dbo.FormiranjeOznake(' Gabon ');
```

T-SQL

- Primer funkcije – dobavljanje podataka o državama sa zadatim uređenjem

```
CREATE FUNCTION dbo.CitanjeDrzavaPoUredjenju(  
    @uredjenje char(3))  
RETURNS TABLE  
AS  
RETURN  
(  
    SELECT DrzNaziv + ' (' + DrzGlavniGrad + ')' AS 'Drzava'  
    FROM dbo.DRZAVA, dbo.UREDJENJE  
    WHERE DrzUredjenje = UreId AND UreOznaka = @uredjenje  
);  
GO  
  
SELECT * FROM dbo.CitanjeDrzavaPoUredjenju('MON');  
GO
```

T-SQL

- Zadatak za funkcije
 - Napisati funkciju za prebrajanje reči u zadatom nazivu

Literatura

- Microsoft Docs. Transact-SQL Reference (Database Engine) - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/language-reference?view=sql-server-ver15>
- Microsoft Docs. Variables (Transact-SQL) - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/language-elements/variables-transact-sql?view=sql-server-ver15>
- Microsoft Docs. Control-of-Flow - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/language-elements/control-of-flow?view=sql-server-ver15>
- Microsoft Docs. Stored Procedures (Database Engine) - SQL Server. Internet: <https://docs.microsoft.com/en-us/sql/relational-databases/stored-procedures/stored-procedures-database-engine?view=sql-server-ver15>

Literatura

- Microsoft Docs. CREATE PROCEDURE (Transact-SQL) - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/statements/create-procedure-transact-sql?view=sql-server-ver15>
- Microsoft Docs. Cursors (Transact-SQL) - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/language-elements/cursors-transact-sql?view=sql-server-ver15>
- Microsoft Docs. What are the Microsoft SQL database functions? - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/functions/functions?view=sql-server-ver15>

Literatura

- Microsoft Docs. CREATE FUNCTION (Transact-SQL) - SQL Server. Internet:
<https://docs.microsoft.com/en-us/sql/t-sql/statements/create-function-transact-sql?view=sql-server-ver15>